
SYSDIFEQ – Systems of First-order Differential Equations 

© 2019 Valentín Albillo 

 

Abstract 

SYSDIFEQ is a program written in 1980 for the HP-41C programmable calculator to obtain an approximate numerical solution for a 

system of N simultaneous first-order differential equations using a fourth-order Runge-Kutta method. Two worked examples included. 

1. Introduction 

SYSDIFEQ is a 137-step RPN program that I wrote in 1980 for the HP-41C programmable calculator (will also 

run in the HP-41CV/CX and the HP42S with trivial or no changes, see Note 1), to obtain an approximate 

numerical solution (using a 4
th

-order Runge-Kutta method) for a system of N simultaneous first-order 

differential equations of the general form: 
 

y1’ = f1 ( x, y1, y2, ..., yn )    

y2’ = f2 ( x, y1, y2, ..., yn ) 

   ... ... ...   with the initial conditions:  x = x0,  y1 = y1 (x0),  ... ,  yn = yn (x0) 

yn’ = fn ( x, y1, y2, ..., yn )   

 

where the  fi  are user-defined functions of the variables x, y1, y2, ..., yn and y1, y2, ..., yn are functions of x.  
 

We seek to compute the values yi (x0 + h), where h is an arbitrary increment of the independent variable x and 

the index  i = 1, 2, ..., n. The input consists of: 
 

a) the equations:    yi’ = fi ( x, y1, y2, ..., yn ) 

b) the increment:   h       (generally small) 

c) the initial values:  yi (x0) 
 

and the output is  yi (x0 + h) , yi (x0 + 2h), ... , for  i = 1, 2, ..., n.  

To this purpose, the 4
th

-order Runge-Kutta method of Gill is applied as shown in the following pseudocode: 
 

for j = 1 to 4 

 for i = 0 to n 

  ki, j = fi (y0, j-1, y1, j-1, ..., yn, j-1)      where the constants are: 

 next i         

 for i = 0 to n           a1 = c1 = c4 = 1/2  

  yi, j = yi, j-1 + h (aj (ki, j - bj qi, j-1))      a2 = c2 = 1 – 1/√2, a3 = c3 = 1+1/√2 

  qi, j = qi, j-1 + 3 (aj (ki, j - bj qi, j-1)) – cj ki, j      a4=1/6, b1 = 2b2 = 2b3 = b4 = 2 

 next i          

next j 

output  yi  for i = 0 to n 
 

This procedure is repeated as often as desired to yield  yi (x0 + h) ,  yi (x0 + 2h) , ...  

The notation y0 = x is used for notational convenience and to simplify the form of the process.  

Initially qi,0 = 0,  yi,0 = yi (x0 ),  thereafter qi,0 (xi ) = qi,4 (xi-1) 

1.1 N-th order differential equations 

 

A single differential equation of orden n:   

 

y
(n)

 = f (x, y, y’, y’’, ... , y
(n-1)

 ) , subject to initial conditions  x = x0, y = y0, y’ = y’0, ..., y
(n-1) = 

y0
(n-1) 

 

can be reduced to a system of n first-order differential equations and solved using this program. 

 

For example, consider the 3
rd

-order differential equation: 

   

  y’’’ = 6y’’ + 7y’ – 8y – 4x + 7,  subject to initial conditions  x0=1, y(1) = 2, y’(1) = 3, y’’(1) = 4 

 

and substitute:  y = y1,  y’ = y2,  y’’= y3  and thus  y’’’ = y3’ , so the equation may be written as: 

 

  y3’ = 6y3 + 7y2 – 8y1 – 4x + 7 

  y2’ = y3     subject to initial conditions  x0=1, y1(1) = 2, y2(1) = 3, y3(1) = 4 

  y1’ = y2 

 

which is a system of 3 first-order differential equations that can be solved using this program. 

 

Using four memory modules, in theory equations up to the 80
th

 order might be solved because after the last one 

all the others take only 5 bytes each to define. 

 

 

1.2 Program characteristics 

 

This program (labeled as “RK” for Runge-Kutta) is 32-register, 224 bytes long so it exactly fits into a single 

magnetic card as long as it doesn’t have a final END instruction, let the final .END. of program memory do the 

task (else it won’t fit into a single card, a second card would be required.) 

The program solves a system of N first-order differential equations, where N depends on the available memory:  
 
 

Number of modules Number of equations  Average size eq. 

none 2 18 bytes 

1 6 63 bytes 

2 10 74 bytes 

3 15 72 bytes 

4 20 71 bytes 

4 30 41 bytes 

4 40 25 bytes 
 
 

Of course, the more equations in the system the more bytes it will probably take to define each. The program is 

also fast, the loops are designed to run as fast as possible by duplicating indirect addresses. Computing times 

depend mostly on the number and complexity of the functions you define but the following times are typical:  

 

Number of equations 1 2 3 5 

Time per point 12 sec. 22 sec. 32 sec. 62 sec. 

2. Program Listing 

 

 
 

01 ♦LBL “RK”    29  STO 09    57  ISG 14   85  STO 15  113  ST+ 04  

02  CLRG     30  STO 00  58  GTO 01 ►   86  RCL 10  114  ISG 13 

03  FIX 0   31  ENTER↑  59  “INC=?”     87  STO 14      115  GTO 11 ►    

04  CF 29     32  SQRT     60  PROMPT    88 ♦LBL 13  116  SF 29  

05  SF 21   33  LASTX    61  STO 12   89  RCL IND 15    117  GTO 11 ►   

06  “N=?”    34   +    62 ♦LBL 14   90  RCL IND 13 118 ♦LBL 10   

07  PROMPT  35  STO 01  63  .002    91   /      119  RCL 10   

08  STO 04  36   /       64  STO 13    92   3   120  0 
09   3     37  STO 02  65   6    93  FS? 29  121  ADV   
10   *      38  RCL 04       66  STO 04   94  ST* Y  122 ♦LBL 02 

11  22   39  ST- 05   67  STO 03   95  RCL IND 16 123  “Y” 

12   +   40  20         68 ♦LBL 11   96  RCL IND 04 124  FIX 0  

13  STO 05    41   +     69  RCL 10   97   *   125  ARCL X 

14  SF 25    42  ST+ 11   70  STO 14   98  RCL IND 15 126  ├“=”  

15  ARCL IND X 43  1 E3      71  RCL 11   99   -   127  FIX 7 

16   1     44   /    72  STO 15  100  RCL IND 13 128  ARCL IND Y  

17  STO 11  45  20    73  RCL 05  101   /       129  AVIEW      

18  STO 07    46   +     74  STO 16   102  ST- IND 14 130  ISG X 

19  STO 08  47  STO 14  75 ♦LBL 12  103   *     131  X<>Y 

20   +     48  STO 10  76  XEQ IND 14 ► 104   +   132  ISG Y      

21  FS?C 25  49 ♦LBL 01  77  RCL 12  105  ST- IND 16 133  GTO 02 ►   

22  GTO 00 ►  50  “Y”       78   *   106   1   134  GTO 14 ►   
23  “SIZE”    51  ARCL 12  79  STO IND 15 107  ST+ 15  135 ♦LBL 20    
24  ARCL X     52  ├“=?”         80  ISG 15   108  ST+ 16  136   1 

25  PROMPT  53  PROMPT   81  X<>Y   109  ISG 14  137  RTN 

26 ♦LBL 00  54  STO IND 14 82  ISG 14  110  GTO 13 ►  138  .END.   

27   2     55  ISG 12  83  GTO 12 ►  111  FS?C 29       

28  STO 06   56  X<>Y     84  RCL 11  112  GTO 10 ►   

 

 

The symbols ♦ and ► are purely cosmetic, to indicate branching;  ├ is the Append alpha function 

 

 

2.1 Contents of the storage registers 

 

Register Contents Register Contents Register Contents 

00 a1 = 2 10 add.f,y 20 y0 = x 

01 a2 = 3.41+ 11 add.k ... ... 

02 a3 = 0.58+ 12 h N+20 yn 

03 a4 = 6 13 add.a N+21 k0 

04 N, add.b 14 add.f,y var. ... ... 

05 add q = 2N+22 15 add k var. 2N+21 kn 

06 b1 = 2 16 add q var. 2N+22 q0 

07 b2 = 1 17 free ... ... 

08 b3 = 1 18 free 3N+22 qn 

09 b4 = 2 19 free   

3. Usage Instructions 

To solve a system of N first-order differential equations perform the following Steps: 

 

Step 1: in RUN Mode:   GTO .138   →  { switch to PRGM Mode }.  

If there are definitions left from previously solved systems, execute: 

  DEL 999  → { you should see:  137 RTN }     

 

Step 2: still in PRGM Mode, define the equations. Every equation  fi  must be defined under a numerical 

label whose number is i+20. This is:  f1 must be defined under LBL 21,  f7 under LBL 27,  f14 

under LBL 34, and so on. Each definition must be terminated with a RTN instruction. 

 To define each equation, yi  can be found stored in register Ri+20  and the convention  y0 = x  is 

used. This is:  x (= y0 ) can be found in R20,  y1 in R21, ...,  y14 in R34 , and so on. 

 You may use R17, R18 and R19 as scratch registers if you wish but do not disturb the contents of any 

other registers from register R00  up to  R3N+22 , both included. See Examples. 

 

Step 3: once the equations have been defined in program memory, switch back to RUN Mode and execute the 

program, as follows: 

   XEQ  “RK”  →  N=?  { asks for the number of equations in the system} 

 N   R/S    →  Y0=? { remember the convention  y0 = x0 , so it’s asking for x0 } 

x0  R/S    →  Y1=? { asks for  y1 (x0) } 

y1  R/S    →  Y2=? { asks for  y2 (x0) } 

  ... ... ... ... ...  

yn  R/S    →  INC=? { asks for the increment [step] h } 

 

Note: the increment (step) h should be adequately small. The smaller is h, the greater is the accuracy and 

number of stages required to reach a given value of x (and so the longer it will take and the more rounding 

errors will accumulate). The error per stage is approximately proportional to h
5
 so as a rule of thumb h=0.1 

should give about 5 correct places at the end of 10 stages or so. 

 

h  R/S    →   { computation begins } 

 

After some time, the values of  yi (x0 + h) will be output. If there’s a printer attached and On, all 

values will be printed and spaced and computation will proceed to the following stage automatically. 

Otherwise, simply press  R/S  after each displayed value to see the next one or resume the 

procedure, like this: 

 

→  Y0=its value { remember, y0 = x0 , so this is x = x0 + h in this stage} 

  R/S    →  Y1=its value { outputs  y1 (x0 + h) } 

 R/S    →  Y2=its value { outputs  y2 (x0 + h) } 

  ... ... ... ... ...  

  R/S    →  YN=its value { outputs  yn (x0 + h), the last value in this stage } 

 R/S    →  Y0=its value { proceeds to compute the next stage, yi (x0 + 2h) } 

 

and so on. To stop the procedure at any time, press  R/S . For another system, go to Step 1. 

Notes: 

 

 if after introducing N a message  SIZE nnn  does show up this means the current register allocation is 

insufficient to run the program, so simply execute  SIZE  nnn  and then press  R/S  to resume. 

In general, if your system has N equations, a minimum SIZE of 3N+23 registers is required. 

 

 do not disturb the stack while the output is taking place. If you want to simply view (not write down) 

each value, without having to press  R/S  each time to continue (you have no printer attached), then 

at any moment stop the computation by pressing  R/S  and execute  CF 21  , then  R/S  again to 

resume. Subsequent values will be displayed without stopping program execution. 

 

 if you want to change the spacing (step) h during the process, wait for the program to stop or output 

something, then stop (if not yet halted) and input the new h like this: 

 

new value of h  STO 12    R↓    R/S  . 
 

    the procedure will resume from where it left but this time using the new h. 
 

 to change the number of decimals in the output, change line 127 FIX 7 to the desired display setting. 

 

 

 

4. Examples 

The following examples can be useful to check that the program is correctly entered and to understand its usage. 
 
 
 

4.1 Example 1 

Solve the system:   

 

 

y1’ = sin x – y2         x0 = 0.230253487 

y2’ = e 
x
 + y3      with initial conditions:  y1 = -0.258919089  

y3’ = 1 – y1 – y2        y2 = 1.487143417 

                                              y3 = 0.973608574 

 

 

using a step h = -0.102342187. 

 

 

Define the equations:  in RUN Mode,   GTO .138  , switch to PRGM Mode,  DEL 999  → see:  137  RTN   

Enter the definitions below, and then switch to RUN Mode, execute  PACK  ,  RAD  and run the program:   

    

138 ♦LBL 21 144 ♦LBL 22 150 ♦LBL 23         XEQ  ”RK”  →  N=?  

139  RCL 20 145  RCL 20 151   1            3  R/S  →  Y0=? 

140  SIN 146  E↑X 152  RCL 21  0.230253487   R/S  →  Y1=? 

141  RCL 22 147  RCL 23 153   - -0.258919089  R/S  →  Y2=? 

142   - 148   + 154  RCL 22  1.487143417  R/S  →  Y3=? 

143  RTN 149  RTN 155   -  0.973608574  R/S  →  INC=? 

  156  RTN -0.102342187  R/S  →  computing proceeds and then ... 

... the output begins →  Y0=0.1279113  { remember, y0 = x0 , so this is x = x0 + h in this first stage} 

   R/S   →  Y1=-0.1364522  { outputs  y1 (x0 + h) }  

 R/S   →  Y2=1.2640152   { outputs  y2 (x0 + h) } 

   R/S   →  Y3=0.9918304   { outputs  y3 (x0 + h), the last value in this stage } 

 R/S   →  Y0=0.0255691  { this is x = x0 + 2h in this second stage} 

   R/S   →  Y1=-0.0258989  { outputs  y1 (x0 + 2h) }  

 R/S   →  Y2=1.0514656  { outputs  y2 (x0 + 2h) } 

   R/S   →  Y3=0.9996729   { outputs  y3 (x0 + 2h), the last value in this stage, and so on ... } 

To check the accuracy obtained, the exact solution is: 

  

 y1 = 1 - e 
x
        y1 (x0 + h) = -0.136452195 

 y2 = e 
x
 + sin x    thus:   y2 (x0 + h) = 1.264014981  ,  so we got 7-8 correct places 

 y3 = cos x      y3 (x0 + h) = 0.991830497 

 

          

4.2 Example 2 

Solve the system: 
 

 

y1’ = y1 – y2 + e 
x
 – y4 – x         x0 = 0 

y2’ = y1 – sin x + e 
x
          y1 = 1 

y3’ = cos x – y3 – y4 – x    with initial conditions:  y2 = 1 

y4’ = y3 – e
 - x

 – 1         y3 = 2 

y5’ = (y5 + sin x – y4)
2        

y4 = y5 = 0 
 

 

Using a step h = 0.1,  find the values of  y1 .. y5  for  x = 1. 

 

 

Define the equations:  in RUN Mode,   GTO .138  , switch to PRGM Mode,  DEL 999  → see:  137  RTN  

Enter the 5 definitions below, and then switch to RUN Mode, execute  PACK  ,  RAD  and run the program:  

[*] The example assumes there were less than 38 registers allocated so it prompts  

   for the user to allocate them right now, before resuming. 

      

138 ♦LBL 21 150 ♦LBL 22 159 ♦LBL 23 169 ♦LBL 24 178 ♦LBL 25    XEQ  ”RK”  →  N=?  

139  RCL 21 151  RCL 21 160  RCL 20 170  RCL 23 179  RCL 25     5  R/S  →  SIZE 38  
[*]
 

140  RCL 22 152  RCL 20 161  COS 171  RCL 20 180  RCL 20 SIZE 38  R/S  →  Y0=? 

141   - 153  SIN 162  RCL 23 172  CHS 181  SIN     0   R/S  →  Y1=? 

142  RCL 20 154   - 163   - 173  E↑X 182   +     1  R/S  →  Y2=? 

143  E↑X 155  RCL 20 164  RCL 24 174   - 183  RCL 24     1  R/S  →  Y3=? 

144   + 156  E↑X 165   - 175   1 184   -     2  R/S  →  Y4=? 

145  RCL 24 157   + 166  RCL 20 176   - 185  X↑2     0  R/S  →  Y5=? 

146   - 158  RTN 167   - 177  RTN 186  RTN     0  R/S  →  INC=? 

147  RCL 20  168  RTN      0.1  R/S  →  computation 

148   -           proceeds  

149  RTN           and then ... 

... the output begins →  Y0=0.1   { remember, y0 = x0 , so this is x = x0 + h = 0.1 in this first stage} 

   R/S   →  Y1=1.0948375  { outputs  y1 (0.1) }  

 R/S   →  Y2=1.2050044   { outputs  y2 (0.1) } 

   R/S   →  Y3=1.8998416  { outputs  y3 (0.1) }  

 R/S   →  Y4=-0.0001665   { outputs  y4 (0.1) } 

   R/S   →  Y5=0.0003345  { outputs  y5 (0.1), the last value in this stage } 

 R/S   →    ...   { proceeds to compute the next stage, yi (0.2) } 

 

After 10 such stages you should get: 

 

 

 x = 1.0000000 

 y1 = 1.3817719        y1 = sin x + cos x 

 y2 = 3.5597527    and the exact solution is:  y2 = sin x + e 
x
   so we got 5-6 correct places 

 y3 = 0.9081817        y3 = cos x + e 
- x

 

 y4 = -0.1585284       y4 = sin x – x 

 y5 = 0.5573977        y5 = tan x – x  

 

 

 

 

Notes 

1.This program will also run in the HP42S but as it has a wider numeric range and greater precision (12-digit instead of 
10-digit), the results might differ slightly. Consult the HP42S Owner’s Manual for other minor differences (e.g.: naming). 

2.This program was published in PPC Melbourne Chapter Technical Notes V1N2 pp23-28  (September 1980). However, 
there’s a missing minus sign (“–“) between e 

x
 and  y4  in the definition of the first equation of the second example. 

3. It was also published in PPC Calculator Journal V8N1 pp17-19  (Jan./Feb. 1981). However, there’s also a missing minus 
sign (“–“) between e 

x
 and  y4  in the definition of the first equation of the second example. 

